



ÚVOD DO SPICE

pomocí programu Micro-Cap

UČEBNÍ TEXTY

Zdeněk Bolek

OBSAH

1. SPICE – CO JE TO?.....	3
2. SPICE – JAK TO FUNGUJE?	4
3. ÚVODNÍ PŘÍKLAD	5
4. NEJPOUŽÍVANĚJŠÍ OBVODOVÉ PRVKY	7
PASIVNÍ PRVKY (R, L, C).....	7
POLOVODIČOVÉ PRVKY	8
ZDROJE NEZÁVISLÉHO NAPĚTÍ A PROUDU.....	9
5. PODOBVODY SPICE	13
DEKLARACE PODOBVODU	13
POUŽITÍ (VOLÁNÍ) PODOBVODU.....	14
STRUKTURA NETLISTU S PODOBVODEM.....	14
PŘÍKLAD ZAPOJENÍ S PODOBVODY.....	15
GLOBÁLNÍ A LOKÁLNÍ UZLY	15
PODOBVODY NA INTERNETU	16
6. DALŠÍ UŽITEČNÉ FUNKCE.....	18
PROMĚNNÉ.....	18
TEPLOTNÍ ZÁVISLOST ODPORU	18
7. VYŘEŠENÁ ÚLOHA	20
8. PŘÍLOHY	24
A. CO JE TO NAPĚŤOVÝ UZEL?	24
B. TYPOGRAFICKÉ KONVENCE.....	26
C. SPOLUPRÁCE SPICE - MICRO-CAP.....	27
D. HLAVNÍ TYPY ANALÝZ.....	30
LITERATURA.....	31

1. SPICE – CO JE TO?

SPICE je zkratka z anglického „Simulation Program with Integrated Circuits Emphasis“, což volně přeloženo znamená „Simulační program použitelný zejména pro integrované obvody“.

SPICE označuje určitou třídu programů pro simulaci činnosti elektrických obvodů, ale také formát dat, se kterými tyto programy pracují.

Mezi simulační programy standardu SPICE dnes patří např. PSPICE, CADENCE-SPICE, H SPICE, SmartSPICE a další. S datovým formátem SPICE je kompatibilní řada dalších simulátorů jako je např. Micro-Cap. Simulační program načítá data popisující elektrické zapojení a podle pokynů provádí požadované typy analýz.

Data bývají soustředěna do textového souboru zvaného *netlist*. Tento soubor obsahuje úplný popis obvodu a příkazy pro simulátor.

Část netlistu popisující elektrický obvod se řídí jednoduchými pravidly, která v mnohém připomínají některé programovací techniky. Jde o to popsat pomocí textového souboru ty vlastnosti elektrického zapojení, které jsou podstatné pro jeho chování. Jakkoli složité zapojení obsahující libovolné prvky tak může být popsáno jediným textovým souborem, po jehož načtení může simulátor předpovědět, jak se toto zapojení bude chovat ve skutečnosti.

Existuje několik důvodů, proč se seznámit s formátem SPICE. Ačkoliv je předurčen k popisu složitých objektů, je to formát jednoduchý a navíc textový. Občasná práce s textovou podobou zapojení má tu výhodu, že jsme nuceni soustředit se jen na to, co je nezbytné pro funkčnost, což nás přímo vede ke správnému „obvodářskému“ myšlení i při čtení schémat v klasické grafické podobě. Kromě toho je netlist SPICE celosvětově uznávaným standardem. V tomto formátu publikují všichni významní výrobci integrovaných obvodů vnitřní zapojení svých součástek, takže si je můžeme po stažení z Internetu sami otestovat na počítači. Nejde přitom pouze o polovodičové součástky, zájemce si může najít např. knihovny běžně vyráběných vakuových elektronek a předběžně provést řadu počítačových pokusů na svém budoucím elektronkovém zesilovači. Textový charakter formátu SPICE je navíc pro manipulaci přes Internet velmi výhodný díky výbornému kompresnímu poměru, takže knihovny součástek zabírají po „zabalení“ málo místa. A nakonec jeden argument ve prospěch „bastlů“ - export do simulačního programu přes formát SPICE může být mnohdy jedinou rozumnou možností, jak si odsimulovat činnost zapojení nakresleného v některém editoru pro návrh plošných spojů (systémy typu EAGLE apod.).

Pro plné využití učebního textu se předpokládá, že čtenář má nainstalovány aktuální verzi obvodového simulátoru Micro-Cap nebo jeho studentskou verzi, která je zdarma k dispozici na internetových stránkách www.spectrum-soft.com, a umí s ním pracovat. Základy SPICE je samozřejmě možno „trénovat“ na jakémkoli simulátoru kompatibilním s tímto formátem.

Otázky ke kapitole 1:

- Co znamená zkratka SPICE a co se jí dnes označuje v elektrotechnice?
- Čím je charakteristický datový formát SPICE?
- Z jakých důvodů je dobré ovládat formát SPICE?

Tipy a otázky pro další studium:

Historie SPICE je stručně popsána např. na stránkách


<http://www.ecircuitcenter.com/SpiceTopics/History.htm>,

http://www.allaboutcircuits.com/vol_5/chpt_7/2.html .

- Jak se jmenoval předchůdce SPICE, kde a kdy vznikl?
- V jakém programovacím jazyku byl napsán první simulátor SPICE?
- Čím jsou charakteristické hlavní verze SPICE, kdy došlo k přepsání do jazyka C?

2. SPICE – JAK TO FUNGUJE?

Netlist se načte do simulátoru, který tak dostane data o elektrickém obvodu a příkazy, co má s těmito daty udělat.



Obr. 1: Spolupráce mezi netlistem a simulátorem

Z obr. 1 je možno vyčíst obecnou strukturu netlistu ve formátu SPICE.

Hlavička je povinný řádek, který slouží pouze pro dokumentační účely a nemá vliv na průběh simulace.

Popis obvodu je část netlistu, ve které jsou popsány vlastnosti všech součástek analyzovaného obvodu a způsob jejich vzájemného propojení.

Příkazy pro řízení simulace říkají simulátoru, který typ analýzy se má spustit, s jakými parametry a co se má s výsledky udělat.

Netlist je ukončen příkazem **.end**.

V této struktuře mají své pevné místo pouze hlavička a ukončovací příkaz „.end“. Řádky obsahující popisná data a příkazy pro simulátor mohou být mezi sebou libovolně „promíchány“. Pro lepší čitelnost netlistu se však doporučuje zachovávat přehlednou strukturu podle obr. 1.

Všechny SPICE-kompatibilní simulátory čtou příkazy pro řízení simulace z netlistu. Většině simulátorů je navíc možno zadat tyto údaje pohodlně přes grafické rozhraní, do kterého se údaje z netlistu načítají pouze jako editovatelné položky. V některých případech je proto možno řídicí příkazy z netlistu úplně vynechat, jak je tomu např. u simulačního programu Micro-Cap. Řídicí sekce netlistu se pak generuje automaticky až po ukončení první simulace, po každém dalším simulačním běhu se sama aktualizuje (viz obr. 1 a Příloha C).


Z výše uvedených důvodů se v tomto učebním textu příkazům pro řízení simulace nebudeme věnovat. Netlisty, se kterými budeme pracovat, budou obsahovat pouze hlavičku, popis zapojení a ukončovací příkaz.

Otázky ke kapitole 2:

- Co je to netlist SPICE a jaká je jeho struktura?
- Jak funguje spolupráce netlistu se simulátorem?
- Proč není pro úspěšnou práci s některými simulátory nutné znát příkazy pro řízení simulace?

3. ÚVODNÍ PŘÍKLAD

Dejme tomu, že bychom chtěli provádět pokusy se zapojením podle obr. 2. Na volné svorky bychom nejprve připojili zdroj stejnosměrného napětí (např. plochou baterii 4,5V) a zajímalo by nás, jakým způsobem se budou po připojení zdroje ustalovat napětí a proudy.




Obr. 2: Příklad skutečného obvodu

Schéma zapojení včetně stejnosměrného zdroje napětí je na obr. 3.

Simulátory používají pro výpočty toho, co se v obvodu děje, modifikovanou metodu uzlových napětí. Pro přepis do textového formátu SPICE tedy musíme nejprve ve schématu vyznačit a pojmenovat *napěťové uzly*. Podrobnosti o tom, co to je uzel a jak se dá pojmenovat, jsou v Příloze A. V naší ukázce jsou za jména uzlů zvolena čísla 1, 2 a 3. Každý obvod *musí* obsahovat referenční uzel 0 (zem), vůči kterému se vztahují napětí ostatních uzlů.

Na obr. 4 je textová podoba zapojení podle obr. 3. První řádek je povinná hlavička, poslední řádek obsahuje ukončovací příkaz. Řádky začínající hvězdičkou (*) jsou komentáře a vše ostatní je popis vlastního zapojení.



Obr. 3: Schéma zapojení s vyznačením uzlů

Obr. 4: Netlist SPICE

Znak „+“ na začátku řádku znamená, že tento řádek je pokračováním řádku předchozího. Interpret SPICE jej vyhodnotí jako znak pro sčítání textových řetězců. Netlist může být těmito značkami udržován v čitelné podobě i v případech, kdy příkaz přesáhne svou délkou obvyklou mez.

Netlist uchováváme v souboru, odkud si jej načte simulátor. Obvyklé přípony těchto textových souborů jsou CIR (circuit), LIB (library), CKT.

Zapojení podle obr. 3 se dá slovně popsat takto (sledujte přitom odpovídající řádky souboru SPICE na obr. 4):

- Napěťový zdroj V_{zdroj} je připojen kladnou a zápornou svorkou na uzly 1 a zem a má hodnotu napětí 4,5V.
- Induktor L_s je připojen mezi uzly 1 a 2 a má indukčnost 10mH.
- Rezistor R_s je připojen mezi uzly 2 a 3 a má odpor 1kW.
- Rezistor R_p je připojen mezi uzel 3 a zem a má odpor 3,3kW.
- Kapacitor C_p je připojen mezi uzel 3 a zem a má kapacitu 220nF.

Podrobnosti o formátech číselných konstant jsou uvedeny v Příloze B.

Postup, jak výše uvedený netlist zpracovat v simulátoru Micro-Cap, je popsán v Příloze C.

V naší ukázce figurují pouze pasivní RLC prvky a zdroj konstantního napětí. Formátem SPICE však lze popsat jakkoliv složité zapojení s libovolnými prvky, přičemž obecný princip zápisu je stejný jako v ukázce: na jednotlivých řádcích netlistu se vyjmenují všechny prvky schématu, přičemž struktura řádku se řídí uskupením

- jméno prvku
- seznam uzlů, ke kterým jsou připojeny jeho vývody
- vlastnosti prvku.

Ve SPICE tvoříme zapojení ze základních prvků (viz kap. 4) a z tzv. podobvodů (viz kap. 5).

Mezi základní prvky patří pasivní (RLC prvky, zpožďovací linky) a polovodičové součástky (diody, tranzistory všeho druhu), zdroje napětí a proudu (nezávislé i řízené) a jiné (nelineární magnetické materiály, spínací prvky aj.).

Podobvod je součástka, kterou si vytvoří uživatel zapouzdřením složitějšího celku, takže ji může zapojovat do obvodu jako jeden prvek. Jako podobvody jsou např. modelovány celé čipy integrovaných obvodů (IO). Získáním takového modelu (např. od výrobce IO na Internetu) dostáváme možnost přímo si odzkoušet předpokládané chování součástky na simulátoru. Pomocí podobvodů SPICE si tak můžeme doplňovat knihovny simulátoru o modely dalších součástek.

Otázky ke kapitole 3 (viz také Přílohy A a B):

- Jaký je postup při vytváření netlistu, který má odpovídat skutečnému zapojení?
- Co je to napěťový uzel a proč se v netlistu nesmí zapomenout na uzel 0?
- Jak se do netlistu promítne zapojení každé jednotlivé součástky (název, vývody, vlastnosti)?
- Jak by se popsaly v netlistu SPICE vlastnosti rezistoru 4M7 a kapacitoru M33?
- Proč je logické, že vůbec nezáleží na pořadí, ve kterém do netlistu zapisujeme jednotlivé prvky obvodu?
- Jaký je rozdíl mezi základním prvkem a podobvodem SPICE?

4. NEJPOUŽÍVANĚJŠÍ OBVODOVÉ PRVKY

Každá součástka je popsána 1 řádkem textu, který se řídí syntaxí

<jméno součástky> <seznam uzlů, na které vedou její vývody> <její vlastnosti>

Typ součástky je dán prvním znakem jejího jména (např. R = rezistor, L = induktor, C = kapacitor atd.), dalšími znaky (xx..xx) se jméno doplní tak, aby bylo v celém zapojení jedinečné.

Pozn.: Jméno by nemělo být doplňováno automaticky např. nic neříkajícími indexy 1, 2, ..., ale mělo by vyjadřovat účel součástky v obvodu (tak např. sériový odpor by se nemusel jmenovat R1, ale Rs). Výrazně se tím zlepší naše orientace v netlistu.

PASIVNÍ PRVKY (R, L, C)

Dvojice uzlů „N⁺ N⁻“, ke kterým je připojen pasivní dvojpól, se sice může zapisovat v libovolném pořadí, avšak za kladný směr proudů a napětí se považuje směr od prvního do druhého uzlu v seznamu, tj. od uzlu N⁺ do uzlu N⁻. Poteče-li např. odporem Rs z následujícího příkladu podle obr. 5 proud směrem z uzlu „konec“ do uzlu „spoj“, simulátor předá vypočítanou hodnotu proudu I(Rs) jako zápornou.

Vlastnosti pasivních prvků se obvykle vyjadřují pouze jedním parametrem (odpor, kapacita, indukčnost) v základních jednotkách (ohm, farad, henry) nebo pomocí dohodnutých násobků (viz Příloha B).


Rezistor

Syntaxe:

Rxx..xx <N⁺> <N⁻> <odpor>

Příklad:

Rs spoj konec 1k



Obr. 5: Rezistor


Kapacitor

Syntaxe:

Cxx..xx <N⁺> <N⁻> <kapacita>

Příklad:

Cp konec 0 220nF



Obr. 6: Kapacitor


Induktor

Syntaxe:

Lxx..xx <N⁺> <N⁻> <indukčnost>

Příklad:

Ls vstup spoj 10mH



Obr. 7: Induktor

POLOVODIČOVÉ PRVKY

U prvků typu diody nebo tranzistoru už bude záležet na tom, v jakém pořadí zapojíme jednotlivé vývody součástky do obvodu (např. prohozením anody a katody polovodičové diody se může změnit funkce celého obvodu). V netlistu tomu bude odpovídat pevně stanovené pořadí v seznamu uzlů, ke kterým jsou jednotlivé elektrody připojeny.

Vlastnosti polovodičových prvků se vyjadřují více než jedním parametrem. Bývá to celý „vektor“ parametrů, který se popisuje tzv. modelem součástky. Např. model bipolárního tranzistoru obsahuje přes 40 parametrů popisujících jeho chování za nejrůznějších okolností (odpory přechodů, substrátové proudy, nejrůznější „bety“, teplotní závislosti apod.). Vlastnosti konkrétní součástky budou v zápisu SPICE zastoupeny jménem modelu. Model tohoto jména s výčtem konkrétních parametrů by měl být v netlistu definován.

Modely nejrůznějších součástek je možno získat na Internetu. Způsob definování modelu si ukážeme na následujícím modelu LED diody:

```
.model ledka D (N=1.7 RS=.7 CJO=23.9P IS=85.3p BV=6 IBV=10U VJ=0.75 TT=4.32U)
```

Jde o příkaz pro simulátor, proto začíná tečkou. Následuje jméno modelu „ledka“ a kód modelu „D“, což značí, že se jedná o polovodičovou diodu. Následují kulaté závorky se seznamem parametrů charakterizujících tuto diodu. Pro zajímavost – RS je sériový odpor, IS je saturační (zbytkový) proud, TT je časové zpoždění pro průchod signálu.

*Pozn.: Pro každý parametr modelu má SPICE přednastavenou **implicitní** hodnotu. Příkaz*

```
.model ledka D ( )
```

by deklaroval chování takové diody, která by odpovídala „průměrné“ diodě z hlediska SPICE. Do závorky se wpisují jen ty parametry, které se liší od implicitních hodnot SPICE.

A nyní příklad modelu tranzistoru KC509:

```
.model KC509 NPN (BF=1.50K BR=2 IS=12.0P CJC=2.80P CJE=1.40P RC=1.01U VAF=100
+ TF=718.4P TR=10N MJC=375.7M VJC=690.0M MJE=375.77M VJE=690.00M NF=1.259
+ ISE=33.7P ISC=0.001566F IKF=9.44M IKR=4.62K NE=2.18 RE=32.4M VTF=113.5
+ ITF=533.9K XTF=19.25M)
```

Kód součástky je tentokrát NPN (typ vodivosti). Opět pro zajímavost – BF je tzv. dopředná beta, CJC je kapacita přechodu báze – kolektor.

Obecná syntaxe modelu vypadá takto:

```
.model <jméno modelu> <kód modelu> (<parametry odlišné od implicitních>)
```

A nyní k samotnému zápisu jednotlivých aktivních prvků:


Dioda

Syntaxe:

```
Dxx.xx <anoda> <katoda> <jméno modelu>
```

Příklad:

```
Dled zatez kol ledka
```

Bipolární tranzistorSyntaxe:**Qxx.xx <kolektor> <báze> <emitor>` <jméno modelu>**Příklad:**Qled kol buzeni 0 KC509**Příklad obvodu pro indikaci pomocí LED:

```

LED indikace
*
*zapojeni
*
Rk plus zatez 100
Dled zatez kol ledka
Qled kol buzeni 0 KC509
*
*modely soucastek
*
.model ledka D (N=1.7 RS=.7 CJO=23.9P
+ IS=85.3p BV=6 IBV=10U VJ=0.75 TT=4.32U)
.model KC509 NPN (BF=1.50K BR=2 IS=12.0P
+ CJC=2.80P CJE=1.40P RC=1.01U VAF=100
+ TF=718.4P TR=10N MJC=375.7M VJC=690.0M
+ MJE=375.77M VJE=690.00M NF=1.259
+ ISE=33.7P ISC=0.001566F IKF=9.44M
+ IKR=4.62K NE=2.18 RE=32.4M VTF=113.5
+ ITF=533.9K XTF=19.25M)
.end


```

ZDROJE NEZÁVISLÉHO NAPĚTÍ A PROUDUSyntaxe:**Vxx.xx <plus> <minus> <vlastnosti>** pro zdroj napětí,**Ixx.xx <plus> <minus> <vlastnosti>** pro zdroj proudu.

Touto syntaxí lze modelovat

- rozdílné chování jednoho zdroje při běhu různých analýz,
- různé tvary časových průběhů.

Obecná syntaxe vlastností je rozdělena do 3 sekcí (údaje v hranatých závorkách jsou nepovinné)


[[DC] <hodnota>] [AC <amplituda> [<fáze>]] [<definice časového průběhu>]**Obr. 9: Obecná definice nezávislého zdroje**Následují příklady, jak definovat různé časové průběhy (sekce **f**).

Zdroj periodických impulzů

PULSE <h1> <h2> [<Td> [<Tr> [<Tf> [<pw> [<per>]]]]],

h1 a **h2** jsou počáteční a koncové hodnoty napětí nebo proudu,

Td, **Tr**, **Tf**, **pw** a **per** jsou zpoždění do začátku náběžné hrany, šířka náběžné a sestupné hrany, šířka impulzu a perioda opakování (vše v sekundách).



Obr. 10. Význam parametrů zdroje impulzů

Příklady:

Vttl vstup 0 PULSE 0 5V 500u 0 0 500u 1m

ideální obdélníkový TTL signál 1kHz se střídou 1:1,

V1 In 0 PULSE -10 10 0 0.5s 0.5s 0 1s

symetrický trojúhelníkový signál 1Hz od -10V do +10V.

Zdroj sinusového signálu

SIN <offs> <A> [<f> [<Td> [<df> [<ph>]]]]],


offs je střední hodnota, kolem které sinusová veličina kmitá,

A a **f** jsou amplituda a frekvence,

Td je časové zpoždění do náběhu signálu (v sekundách),

df je činitel tlumení v sec^{-1} pro výrobu tlumených nebo narůstajících kmitů a

ph je počáteční fáze ve stupních.



Obr. 11. Význam parametrů sinusového zdroje

Příklady:

Vin vstup 0 SIN 10mV 5mV 1k

sinusový napěťový zdroj 1kHz s amplitudou 5mV kmitající kolem 10mV,

V2 in1 in2 SIN 0 325V 50Hz 0s 0 90


sinusový napěťový zdroj 50Hz s amplitudou 325V kmitající kolem 0 s počáteční fází 90°.

Zdroj po částech lineárního signálu

PWL (t1 h1 t2 h2 .. tn hn),

t1 .. tn je rostoucí posloupnost časů v sekundách,

h1 .. hn jsou hodnoty napětí nebo proudů zdroje v odpovídajících časech t1 .. tn.



Obr. 12. Význam parametrů zdroje PWL

Příklad:

Vstup buzení 0 PWL (0ms 1V 1ms 1V 2ms 2V 3ms 2V 4ms 3V 5ms 3V)

stupňovitě rostoucí signál 0 – 3V.

Obecná syntaxe nezávislého zdroje (viz obr. 9) je rozdělena do 3 sekcí. Sekce • začíná (nepovinným) polem DC, sekce , polem AC a sekce f jedním z výrazů pro časový průběh signálu (např. SIN, PULSE nebo PWL). Z těchto 3 sekcí použijeme při deklaraci zdroje ty, které odpovídají jednému typu zdroje a jedné požadované analýze. Praktické příklady, jak používat obecnou syntaxi zdroje, jsou uvedeny v následující tabulce.

Příklady požadavků na zdroj a analýzu	Použité sekce z obr. 9, příklad
Časová analýza se zdrojem konstantního napětí nebo proudu	• Vbat 1 2 9V
Časová analýza se zdrojem konstantního napětí + frekvenční analýza vůči vstupu, ke kterému je připojen tento zdroj	• , Vin 1 2 4.5V AC 1V
Časová analýza se zdrojem proudu definovaného průběhu + frekvenční analýza vůči vstupu, ke kterému je připojen tento zdroj	, f Is In 0 sin 0 10mA 1kHz AC 1A

Otázky a úkoly ke kapitole 4:

- Popište syntaxi SPICE pro zapojení RLC prvků do obvodu.
- Vytvořte netlist popisující zapojení Wienova článku.
- Popište syntaxi SPICE pro zapojení polovodičové diody a bipolárního tranzistoru.
- Vysvětlete úlohu příkazu „.model“ a jeho strukturu.
- Vytvořte netlist popisující jednotranzistorový zesilovač v základním zapojení se společným emitorem (nastavení báze pomocí odporovým děličem, bez teplotní stabilizace, vstupní uzel je od báze oddělen vazebním kondenzátorem).
- Popište a vysvětlete obecnou syntaxi zdroje nezávislého napětí a proudu.
- Popište ve formátu SPICE zdroj síťového napětí $230V_{ef}/50Hz$ (výhradně pro účel časové analýzy).
- Popište ve formátu SPICE zdroj ideálního obdélníkového napětí TTL o frekvenci 10kHz se střídou 1:1 (pro účely časové a frekvenční analýzy, ve frekvenční analýze se tento zdroj nahradí sinusovým zdrojem o amplitudě 1V).
- Popište ve formátu SPICE zdroj jednorázového trojúhelníkového impulzu rovnoměrně stoupajícího od 0 do 100mV po dobu 10ms a rovnoměrně klesajícího zpět k 0 po dobu 5ms (výhradně pro účel časové analýzy).
- Popište ve formátu SPICE zdroj stejnosměrného napětí 12V s šumovou harmonickou složkou o amplitudě 12mV a frekvenci 120MHz (výhradně pro účel časové analýzy).

Tipy a otázky pro další studium:

Převod schématu zapojení do netlistu SPICE je jednoznačnou úlohou. Získání netlistu z grafické podoby schématu patří mezi základní vybavu většiny simulátorů. V prostředí Micro-Capu jej lze uskutečnit přes nabídku „File → Translate → Schematic to SPICE Text File“.

- Prozkoumejte tuto možnost v prostředí programu Micro-Cap. V čem se liší výsledný netlist pro různé varianty formátu SPICE?

Zajímavá je otázka zpětné rekonstrukce grafické podoby schématu z textového netlistu. Tato rekonstrukce totiž není jednoznačnou úlohou.

- Úkol pro dvojici: každý z dvojice vytvoří netlist popisující Graetzův usměrňovací můstek napájený zdrojem sinusového napětí $24V_{ef}/50Hz$ a předá ho svému protějšku. Oba se pak pokusí zpětně rekonstruovat grafickou podobu zapojení včetně původního popisu uzlů a součástek.
- Zamyslete se nad tím, co všechno by musel zvládnout program, který by načetl netlist SPICE a vytvořil by z něj schéma zapojení.
- Pokuste se najít takový program na Internetu (pozn.: takový program existuje!).

Následující problematiku lze nastudovat pomocí helpu programu Micro-Cap nebo také v literatuře [1]. O SPICE je mnoho zajímavého také na francouzských stránkách

http://lewebelectronique.free.fr/dossiers/spi_indx.htm.

Do jiného jazyka je lze přeložit např. překladačem

<http://translate.google.com/>.

- Jakým způsobem se v syntaxi L a C prvků vyznačí nenulové počáteční podmínky, tzn. počáteční proud cívky a počáteční napětí na kondenzátoru?
- Dá se u RLC prvků použít pro jemnější rozlišení vlastností příkazu „.model“?
- Zjistěte SPICE syntaxi pro modelování cívek se vzájemnými indukčnostmi.
- Zjistěte SPICE syntaxi pro bipolární tranzistor se 4 vývody (vyvedený substrát).
- Zjistěte SPICE syntaxi pro tranzistory JFET, MOSFET.
- Zjistěte SPICE syntaxi pro řízené zdroje napětí a proudu.

Náměty na další studium:

<http://www.ecircuitcenter.com/AboutSPICE.htm>,

<http://newton.ex.ac.uk/teaching/CDHW/Electronics2/userguide/>.

5. PODOBVODY SPICE

Ve SPICE je možno část obvodu pojmenovat a používat ji pod tímto jménem jako nový prvek. Umožňuje to přehlednou práci např. s integrovanými obvody (IO) nebo se složitějšími součástkami, které se ve schématu vícenásobně opakují.

Podobvod (subcircuit) ve SPICE lze přirovnat k podprogramu ve vyšším programovacím jazyce. Podobvod se musí *deklarovat*, tj. pojmenovat a definovat, co všechno je obsaženo v jeho vnitřním zapojení. Pak už se jenom *volá* na místě, kam se do schématu vkládá jako nový prvek. Podobvody lze do sebe vnořovat (stejně jako podprogramy).

DEKLARACE PODOBVODU

Syntaxe:

.SUBCKT <jméno podobvodu> <seznam vývodů>


<popis vnitřního zapojení>

.ENDS

Úvodní příkaz „.subckt“ značí definici podobvodu a závěrečný příkaz „.ends“ znamená konec podobvodu, všechno mezi nimi je popisem vnitřního zapojení. Za klíčovým slovem „.subckt“ následují jméno součástky a seznam pojmenovaných vývodů.

Příklad:

Na obr. 13 je vnitřní zapojení jednoho hradla TTL invertoru.



Obr. 13. Schéma podobvodu TTL invertoru

Šestice těchto invertorů se vyrábí pod označením SN7404. Podobvod podle obr. 13 tedy nazveme SN7404. S okolím bude spojen 4 vývody: „In“ a „Out“ jsou TTL vstup a TTL výstup, „Ucc“ je napájení a „Gnd“ je zem („Gnd“ není žádné klíčové slovo, stejně tak bychom mohli zvolit např. název „Zem“). To jsou vnitřní názvy uzlů. Po deklaraci získáme 4-vývodovou součástku, kterou můžeme zapojovat do sestav s jinými prvky.

```
.subckt SN7404 In Out Ucc Gnd
D1 6 Out DIODA
Q1 2 1 In TRANZISTOR
Q2 3 2 4 TRANZISTOR
Q3 5 3 6 TRANZISTOR
Q4 Out 4 Gnd TRANZISTOR
R1 Ucc 1 4K
R2 Ucc 3 1.4K
R3 4 Gnd 1K
R4 Ucc 5 100
.MODEL DIODA D (IS=10F TT=10N CJO=900F VJ=0.7)
.MODEL TRANZISTOR NPN (BF=75 IS=1F CJC=5P CJE=2P VAF=50 TF=.5N TR=5N
+ VAR=100)
.ends
```

Obr. 14. SPICE zápis podobvodu TTL invertoru

POUŽITÍ (VOLÁNÍ) PODOBVODU

Podobvody jsou považovány za samostatné součástky. Simulátor je pozná podle toho, že jejich jména začínají písmenem „X“. Samotné začlenění do obvodu podléhá úplně stejné syntaxi, jako je tomu u jiných součástek.

Syntaxe:

Xxx.xx <seznam uzlů, ke kterým jsou připojeny vývody podobvodu> <jméno podobvodu>

Příklad:

V deklaraci SN7404 jsou vývody uvedeny v pořadí „In Out Ucc Gnd“. Podle obr. 13 jsou tyto vývody připojeny k uzlům vnějšího obvodu takto:

.subckt SN7404	In	Out	Ucc	Gnd	...vývody podobvodu
	Vstup	Vystup	Plus	0	...uzly vnějšího obvodu

Tomu odpovídá syntaxe

Xinv Vstup Vystup Plus 0 SN7404

STRUKTURA NETLISTU S PODOBVODEM

NETLIST
Hlavička
. . Xinv <vnější uzly> SN7404 . . .subckt SN7404 <vývody> <vnitřní zapojení> .ends .end

NETLIST
Hlavička
. Xinv <vnější uzly> SN7404 . . .lib ./ttl.lib . . .end

SOUBOR „ttl.lib“
. . .subckt SN7404 <vývody> <vnitřní zapojení> .ends . . .

Obr. 15. a) Podobvod v netlistu


Obr. 15. b) Podobvod v externím souboru

Obr. 15 představuje dvě varianty struktury netlistu s podobvodem. U varianty a) je deklarace podobvodu součástí netlistu, u varianty b) je umístěna v externím souboru, na který je v netlistu pouze odkaz příkazem „.lib <úplná specifikace souboru>“. Soubor musí být specifikován jménem i koncovkou, a to buď s celou cestou nebo relativně vzhledem k umístění netlistu. Pokud se netlist i knihovní soubor nacházejí ve stejném adresáři, některé simulátory vyžadují v cestě k souboru značku aktuálního adresáře „./“.

Umísťování podobvodů do externích souborů lze jedině doporučit. Výrazně se tím zpřehledňuje netlist. V jednom externím souboru může být několik podobvodů, což umožňuje vytvářet rozsáhlé knihovní soubory tříděné podle výrobce, typu součástek apod. Způsob, jakým se netlist dostane k datům konkrétního podobvodu, je zřejmý z obr. 15b). Na disku se vyhledá soubor udaný příkazem „.lib“ a v něm se hledá jméno podobvodu v definičních řádcích „.subckt“.

PŘÍKLAD ZAPOJENÍ S PODOBVODY

Na obr. 16 je symetrické zapojení astabilního klopného obvodu se dvěma TTL invertory. Předpokládejme, že model podobvodu je umístěn (ve stejném adresáři jako netlist) v souboru „ttl.lib“ a že obsah tohoto souboru je totožný s výpisem podle obr. 14. Odpovídající netlist je na obr. 17.



Obr. 16. Zapojení AKO



```
Astabilni klopny obvod
*
*Zapis schematu
VZDROJ Ucc 0 5V
R1 In2 0 5.6K
R2 In1 0 4.7K
C1 In2 Out1 100nF
C2 In1 Out2 100nF
XIO1 In1 Out1 Ucc 0 SN7404
XIO2 In2 Out2 Ucc 0 SN7404
*
.lib ./TTL.LIB
*link na knihovnu s podobvodem
.END
```

Obr. 17. Netlist

GLOBÁLNÍ A LOKÁLNÍ UZLY

Přirovnáme-li netlist k hlavnímu programu, podobvody k procedurám a uzly k proměnným, pak lze snadno pochopit, co znamenají globální a lokální uzly.

Globální uzly jsou uzly, se kterými se pracuje přímo na úrovni hlavního netlistu. Na obr. 16 jsou to uzly „In1“, „In2“, „Out1“, „Out2“, „Ucc“ a zemnicí uzel „0“. Lokální uzly jsou ty uzly, které jsou uvnitř podobvodu, v našem případě (viz obr. 13) jsou to „1“ až „6“, „In“, „Out“, „Ucc“ a „Gnd“.

Platí, že označení lokálních uzlů je čistě věcí podobvodu samotného (neboli hlavní netlist dovnitř podobvodu „nevidí“). Z toho vyplývá, že používá-li jeden netlist více různých podobvodů, mohou být v nich použity lokální uzly stejného jména a ničemu to nevadí. To odpovídá běžné situaci, kdy knihovny jsou programovány zcela nezávisle na sobě a my je získáváme z různých zdrojů. Podobvod pouze „zapojíme“ do netlistu jako modul a nemusíme rozumět tomu, jak je uvnitř naprogramován.


Dále platí, že podobvody „vidí“ globální uzly hlavního netlistu, tj. mohou s nimi pracovat. Tak bylo např. možné podobvod podle obr. 14 uzemnit „uvnitř“ na globální uzel „0“ místo propojení na lokální uzel „Gnd“.

V případě výskytu stejného názvu u lokálního i globálního uzlu se nic neděje, hlavní netlist pracuje s globálním a podobvod s lokálním uzlem. Porovnáním obr. 13 a obr. 16 zjistíme, že k této situaci došlo v případě uzlů s označením „Ucc“.

PODOBVODY NA INTERNETU

Používat ke své práci SPICE – kompatibilní simulátor znamená mít možnost vyzkoušet si chování libovolné součástky, jejíž model najdeme na Internetu.

Podobvody SPICE dnes publikují všichni významní výrobci elektronických součástek. Průměrný vyhledávač vrací tisíce odkazů na současný výskyt klíčových slov „spice“ a „subcircuit“. Tyto odkazy většinou vedou na oficiální stránky výrobců součástek, na soukromé stránky těch, kdo s modely pracují a na stránky s nejrůznějším studijním materiálem.



Obr. 18. Výsledek vyhledávání na slova „SPICE“ a „subcircuit“

Podobvody je možno získat jako samostatné soubory nebo bývají součástí textu na stránce. V prvním případě stahujeme soubory, ve druhém musíme vytvořit nový soubor, do něhož umístíme text vybraný např. přes schránku. Můžeme se setkat i s podobvodem uvnitř katalogového listu ve formátu PDF.

Finální soubory s podobvody SPICE nemají unifikovanou koncovku, nicméně nejčastěji se setkáme s příponami LIB (library), CIR (circuit), MOD, MDL (model), TXT (text), FAM (family), SP, SP2, SP3 (SPICE). Soubory bývají často k dispozici jako archivy ZIP nebo jako samorozbalovací archivy EXE.

Některé firmy požadují před stažením souboru registraci. Obvykle je vyžadována pouze e-mailová adresa.

Podobvody SPICE bývají chráněny přes Copyright ©. Mnozí výrobci umísťují krátké licenční ujednání přímo do netlistu jako poznámku. Následující ukázka na obr. 19 pochází z dílny firmy National Semiconductor.

```

* (C) National Semiconductor, Inc.
* Models developed and under copyright by:
* National Semiconductor, Inc.
* ///////////////////////////////////////////////////////////////////
* Legal Notice: This material is intended for free software support.
* The file may be copied, and distributed; however, reselling the
* material is illegal

```

Obr. 19. Ukázka licenčního ujednání v knihovně podobvodů SPICE

Otázky ke kapitole 5:

- Vysvětlete, co je to podobvod SPICE a jaké nové možnosti tento druh součástky nabízí.
- Popište syntaxi definice a volání podobvodu SPICE.
- Najděte na Internetu podobvod některého komerčně vyráběného dynamického reproduktoru a zjistěte si jeho frekvenční parametry a odpor kmitací cívky.
 - Překreslete netlist do grafické podoby a dostanete tak náhradní schéma reproduktoru. Kterým prvkům schématu odpovídají indukčnost a činný odpor kmitací cívky?
 - Použijte podobvod k tomu, abyste proměřili závislost impedance reproduktoru na kmitočtu. Jak lze z této charakteristiky odečíst činný odpor kmitací cívky?

Pozn.: Impedanci lze jednoduše proměřovat tak, že reproduktor budíme harmonickým zdrojem proudu s amplitudou 1A. V AC analýze proladíme frekvenci zdroje v požadovaném rozsahu. Napětí na reproduktoru se pak bude číselně shodovat s jeho impedancí.

Tipy a otázky pro další studium:

Seznamte se s různými zdroji podobvodů SPICE na Internetu. Velké množství internetových odkazů na podobvody SPICE je na adrese

<http://homepages.which.net/~paul.hills/Circuits/Spice/ModelIndex.html>,

seznam je občas aktualizován.

- Jak se liší koncovky souborů s podobvody, které nabízejí firmy z následujícího seznamu?
 - <http://www.intersil.com/design/designmodels.asp>,
 - <http://www.maxim-ic.com/tools/spice/>,
 - <http://www.onsemi.com/site/support/models>,
 - <http://www.analog.com/>
- S jakou překážkou se při získávání podobvodů můžeme často setkat, jak je tomu např. na adrese
 - <http://www.fairchildsemi.com/models/> ?
- Jakou jinou formu nabídky podobvodů zvolily následující firmy?
 - <http://www.metelics.com/spice.htm>,
 - <http://www.diodes.com/spicemodels/>
 - <http://www.coilcraft.com/models.cfm>

Pozn: Se změnou organizace dat na serverech se některé odkazy mohou stát nefunkčními.

6. DALŠÍ UŽITEČNÉ FUNKCE

PROMĚNNÉ

Parametrům součástek jsme zatím přiřazovali konstanty, např. rezistoru jsme zadávali odpor v ohmech apod. Ve formátu SPICE je možno pracovat také s proměnnými. Simulátor může s obsahem takové proměnné pracovat, např. měnit její obsah, což nám otevírá další možnosti.

Syntaxe:

.PARAM <jméno proměnné>=<hodnota>

nebo

.PARAM <jméno proměnné>={<výraz>}

S obsahem proměnné se pracuje tak jako s konstantou, avšak na její místo se dá jméno proměnné ve složených závorkách

{<jméno proměnné>}

Příklad:

.PARAM zatez=100

Rzatez Out 0 {zatez}

Zatěžovací rezistor zapojený mezi výstup „Out“ a zem bude mít odpor 100ohmů, ale pro simulátor je to již proměnná, se kterou umí provádět dodatečné operace. Např. simulátor Micro-Cap umí symbolické proměnné spojitě měnit v zadaném intervalu během simulace. V praxi to znamená např. elegantní možnost, jak realizovat aktivní zátěž, což nám může usnadnit měření zatěžovacích charakteristik.

Příklad:

.PARAM trimr={10k*T}

Rprom 1 2 {trimr}

Rezistor zapojený mezi uzly „1“ a „2“ je proměnný v čase. Pro čas ubíhající mezi 0 a 1s nabývá hodnot 0 až 10kΩ. Ve výrazech se dá využívat všech vnitřních proměnných simulátoru (T je proměnná pro simulační čas).

TEPLOTNÍ ZÁVISLOST ODPORU

U rezistorů se jednoduše zadává teplotní závislost odporu.

Syntaxe:

Rxx.xx <N⁺> <N⁻> <odpor> [TC=<tc1>[,<tc2>]],

tc1 (tc2) je lineární (kvadratický) teplotní koeficient.

Běžně se využívá lineárního teplotního koeficientu, který bývá také uváděn v katalogu v jednotkách „ppm“, což znamená „v miliontinách“ základních jednotek.

Příklad:

R1 NODEa NODEb 100 TC=3m

znamená rezistor s odporem 100 ohmů zapojený mezi uzly „NODEa“ a „NODEb“ s lineárním teplotním koeficientem $3 \cdot 10^{-3}/^{\circ}\text{C}$. Význam tohoto čísla je takový, že při změně teploty od teploty nominální o 1°C nahoru by se zvýšil odpor 1-ohmového rezistoru o $3 \cdot 10^{-3}$ ohmů. U 100-ohmového

rezistoru to bude činit 100x víc, tj. 0,3 ohmů na stupeň. Při zvýšení teploty o 100°C bude celková změna odporu $100 \times 0,3 = 30$ ohmů.

Obecný vzorec závislosti změny odporu na změně teploty je tedy

$$R(v) - R(v_{\text{nom}}) = R(v_{\text{nom}}) \cdot TC1 \cdot (v - v_{\text{nom}}),$$

kde v a v_{nom} jsou aktuální a nominální teploty. Za nominální teplotu se obvykle považuje 300K, což odpovídá 27°C.

Pro $TC1 = 3 \cdot 10^{-3}$ bychom v katalogu našli 3000ppm ($= 3000 \times 10^{-6}$).

Obecně platí

$$PPM = TC1 \cdot 10^6.$$

Pokud není teplotní závislost parametru dostatečně lineární, vyskytuje se v teplotních charakteristikách navíc kvadratický teplotní koeficient TC2.

Pozn.: Výše uvedeným způsobem se definuje teplotní koeficient libovolné veličiny, nejen odporu. Např. teplotní koeficient napětí udává, o kolik voltů by se zvýšila hodnota napětí obdobného zdroje o napětí 1V při zvýšení jeho teploty o 1°C. Jednotkou TK je 1/°C, vzhledem ke zkoumané veličině je to jednotka relativní. Často se používá také absolutní teplotní součinitel, který udává absolutní změnu dané veličiny při zvýšení teploty o 1°C. Jednotkou je pak <jednotka zkoumané veličiny>/°C.

Otázky ke kapitole 6:

- Vysvětlete, co je to proměnná ve SPICE a jaké nové možnosti ohledně simulace nám otevírá.
- Popište způsob, jakým se ve SPICE proměnná definuje a jak se použije.
- Ze dvou rezistorů sestavte napěťový dělič, který se bude chovat jako potenciometr, poloha jezdce bude ovládána parametrem se jménem „poloha“. Změnou parametru od 0 do 1 se bude řídit poloha jezdce od jedné krajní meze k druhé.
- Vysvětlete, co je to teplotní koeficient.
- Vypočítejte, jak se změní hodnota odporu 10kΩ s teplotním koeficientem 1500ppm při jeho přemístění z místnosti vytopené na 25°C do venkovního prostředí o teplotě 5°C.
- Sestavte netlist SPICE popisující napěťový dělič složený ze dvou rezistorů s teplotně závislými odpory. Odsimulujte a zobrazte závislost výstupního napětí na teplotě pro různé kombinace teplotních koeficientů obou odporů. Proč je tato závislost obecně nelineární? Jak je možné, že někdy je teplotní koeficient výstupního napětí kladný a jindy zase záporný? Odvoďte zákonitost, kterou se tento jev řídí.

Tipy a otázky pro další studium:

Nastudujte syntaxi příkazu „model“ pro RLC prvky.

- Jakým univerzálním způsobem se dají simulovat všechny RLC prvky (tj. nejen rezistory) s teplotně závislými parametry?

7. VYŘEŠENÁ ÚLOHA

REGULOVATELNÝ ZDROJ NAPĚTÍ S TL783

Zadání:

- 1) Najděte na Internetu model SPICE a katalogový list napěťového regulátoru TL783 od fy Texas Instruments.
- 2) Podejte stručnou charakteristiku součástky (funkce, přehled parametrů).
- 3) Napište netlist SPICE, který bude popisovat zapojení nastavitelného zdroje napětí 1,25-115V s TL783. Použijte zapojení IO podle doporučení uvedeného v katalogovém listu. Vysvětlete funkci zapojení.
- 4) Provéřte regulaci v uvedeném rozsahu pomocí ovládacího potenciometru. Vytvořte graf znázorňující závislost výstupního napětí na hodnotě odporu potenciometru.
- 5) Nastavte výstupní napětí na 100V při 27 °C a proměřte jeho závislost na teplotě. Změřte teplotní závislost výstupního napětí při 27°C (v jednotkách mV/°C a ppm).

Vypracování:

- 1) Tento podobvod byl nalezen na internetové adrese

http://www.macs.ece.mcgill.ca/~rfic/EC2/SPICE_MODEL_LIB.htm.

```

* TL783C voltage regulator "macromodel"          * CURRENT LIMITING
* subcircuit                                     DSC 6 11 DMOD
*created using Parts release 5.3 on 04/08/93    ESC 11 OUT VALUE {1.96-0.01057*V(6,5)*V(13,5)}
* at 15:09                                       *
* PARTS is a MicroSim product.                  * FOLDBACK CURRENT
*                                                DFB 6 12 DMOD
* connections:      input                    EFB 12 OUT VALUE {2.326 -
*                   | adjustment pin         + 0.03221*V(13,5)+0.0001421*V(13,5)*V(13,5)
*                   | | output              + -0.02*V(13,5)*V(6,5)}
*                   | | |                  EB 7 OUT 8 OUT 8.069
.SUBCKT TL783C      IN ADJ OUT                *
*                                                * ZERO OF OUTPUT IMPEDANCE
* POSITIVE ADJUSTABLE VOLTAGE REGULATOR        RP 9 8 100
JADJ IN ADJ ADJ JADJMOD      ;ADJ. PIN CURRENT CPZ 10 OUT 1.989e-006
VREF 4 ADJ 1.27              DPU 10 OUT DMOD      ;POWER-UP CLAMPLING DIODE
DBK IN 13 DMOD                RZ 8 10 0.1
*                               EP 9 OUT 4 OUT 100
* ZERO OF RIPPLE REJECTION                    RI OUT 4 100MEG
CBC 13 15 8e-010              .MODEL QPASSMOD NPN (IS=30F BF=50 VAF=94.64
RBC 15 5 1000                  + NF=7.604)
QPASS 13 5 OUT QPASSMOD        .MODEL JADJMOD NJF (BETA=8.3e-005 VTO=-1)
RB1 7 6 1                      .MODEL DMOD D (IS=30F N=7.604)
RB2 6 5 85.21                  .ENDS

```

Obr. 20. Výpis podobvodu TL783



Výpis podobvodu regulátoru TL783 je na obr. 20. Jde o soubor se jménem „TL783.lib“.

Z výpisu je patrné, že součástka má 3 vývody s vnitřním pojmenováním „IN“, „ADJ“ a „OUT“. Toto pojmenování a pořadí přívodů je to jediné, co z celého výpisu potřebujeme, abychom dokázali chování součástky odsimulovat. Samotné struktře modelu rozumět nemusíme, pracujeme s podobvodem jako s hotovým modulem.


2) Katalogové listy bývají ve formátu PDF. Z katalogového listu lze získat rychlou informaci o součástce (charakteristika, vzhled pouzdra, zapojení vývodů, aplikační zapojení).

V případě TL783 jde o 3 – svorkový regulovatelný stabilizátor, ze kterého lze odebírat napětí v rozsahu 1,25V až 125V při proudovém zatížení 700mA. Je předurčen pro vysokonapěťové aplikace, ve kterých se nedají použít standardní bipolární technologie.

3) Základní aplikační zapojení TL783 a odpovídající netlist SPICE jsou na obr. 22. Vstupní napětí se přivádí mezi svorku „IN“ a zem, výstupní stabilizované napětí je k dispozici mezi svorkou „OUT“ a zemí. Hodnota výstupního napětí je odvozena z napěťového děliče tvořeného rezistory R1 a R2, přičemž doporučená hodnota R1 je 82Ω a k regulaci slouží R2. Minimální nastavitelná hodnota výstupního napětí $V_{ref} = 1,25V$.



Obr. 21. Pouzdro regulátoru TL783




Obr. 22. Aplikační zapojení a netlist SPICE

Proměnný odpor R2 lze realizovat použitím proměnné „trimr“. Tuto proměnnou můžeme spojitě měnit v DC analýze.

Pozn.: Jiným řešením by bylo zvyšovat R2 rovnoměrně v čase od 0 do maxima. Časový průběh výstupního napětí by pak odpovídal rovnoměrnému zvyšování odporu R2. Růst R2 by musel být dostatečně pomalý, aby se ve výstupním napětí neprojevovala dynamika obvodu (změny by neměly být rychlejší, než je samotný obvod, jinak by byl výsledek zkreslen přechodnými jevy).

4) V souladu s aplikačním zapojením na obr. 22 přivedeme na vstup IO stejnosměrné napětí 125V a provedeme DC analýzu výstupního napětí pro změnu odporu R2 od 0 do 10kΩ. Výsledek je na obr. 23.



Obr. 23. Regulace výstupního napětí TL783


Graf ukazuje srovnání mezi odsimulovaným průběhem a průběhem vypočítaným podle vzorce z katalogového listu. Oba průběhy vycházejí z minimálního napětí regulátoru cca 1,25V a v celé lineární oblasti se prakticky kryjí. U skutečného průběhu se projevuje od hodnoty trimru cca 7,5kΩ omezení výstupního napětí na 115V (regulátor potřebuje pro správnou funkci, aby vstupní napětí převyšovalo výstupní napětí o nezbytnou rezervu).

Další bod předpokládá, že trimrem R2 nastavíme výstupní napětí na 100V. Z grafu na obr. 23 lze vyčíst, že tomuto napětí odpovídá hodnota trimru 6,374kΩ.

5) V netlistu změním hodnotu odporu R2 na 6,374kΩ a tím nastavíme výstupní napětí regulátoru na 100V. Měření teplotní závislosti výstupního napětí provedeme v režimu DC analýzy, která umožňuje měnit teplotu spojitě v daném intervalu.

Měřit budeme v teplotním intervalu od 0°C do 150°C. První průběh z obr. 24 ukazuje závislost výstupního napětí na teplotě. Pro 27°C je na výstupu 100V, s rostoucí teplotou napětí mírně stoupá v celém teplotním rozsahu.

Druhý průběh z obr. 24 ukazuje závislost absolutního teplotního koeficientu výstupního napětí na teplotě. Absolutní teplotní koeficient výstupního napětí je definován jako změna napětí připadající na zvýšení teploty o 1°C. Je to tedy strmost teplotní závislosti a dá se vypočítat jako její derivace podle teploty. Všimněme si, že tato strmost je bod od bodu poněkud jiná, proto změna napětí na 1°C je závislá na konkrétní teplotě. Pro 27°C odečteme z grafu hodnotu asi 1,219mV/°C. Jelikož absolutní velikost výstupního napětí je přitom 100V, na 1V by připadla teplotní změna 12,19μV/°C. Teplotní koeficient je tedy při 27°C asi 12,19 ppm.



Obr. 24. Závislosti výstupního napětí a absolutního teplotního koeficientu na teplotě

Uvedené výsledky vypovídají pouze o teplotních parametrech samotného regulátoru TL783. Teplotní stabilita výstupního napětí cca 12 ppm bude v reálném zapojení znehodnocena mnohem horšími teplotními koeficienty rezistorů R1 a R2.

Tipy a otázky pro další studium:

- Prozkoumejte, jak bude teplotní koeficient (TK, totéž co TC1, viz kap. 6) výstupního napětí ovlivněn TK obou rezistorů R1 a R2. Zamyslete se nad následujícími otázkami:
 - Bude TK výstupního napětí závislý na tom, jak velké napětí bude nastaveno děličem R1 a R2?
 - Dá se jednoduše říct, že s růstem TK obou rezistorů poroste i TK výstupního napětí?

Odpověď hledejte ve vzorci pro výstupní napětí na obr. 22 nebo tento problém vyřešte přímo simulací na počítači. TK odporů R1 a R2 uvažujte v řádu $1e-3$. Při diskusi výsledků pamatujte na to, že pokud je R2 proměnný odpor, bude mít výrazně horší teplotní stabilitu než pevný odpor R1.
- Po připojení zdroje napětí ke vstupu regulátoru TL783 se na výstupu na okamžik objeví napětí vyšší, než kolik jsme nastavili rezistory R1 a R2.
 - Zjistěte, jak lze tento překmit zlikvidovat kondenzátorem připojeným mezi výstupní svorku regulátoru a zem.
 - Zjistěte, jaký vliv má na tento překmit velikost proudu odebíraného z regulátoru.

8. PŘÍLOHY

A. CO JE TO NAPĚŤOVÝ UZEL?


Formát SPICE vychází z toho, že v daném zapojení nejprve vyznačíme všechny *napěťové* uzly. Jejich určení je jednoznačnou úlohou, proto se tato metoda hodí pro počítačové zpracování.

Při troše cviku je určení všech napěťových uzlů snadné. Zásadou je, že nesmíme žádný vynechat ani přidat.


V praxi se nám nestává, že bychom nějaký uzel přehlédli. V tom případě by totiž v hotovém netlistu chyběly všechny součástky, které se do tohoto uzlu sbíhají svými vývody, takže bychom chybu objevili dodatečně. Problémem spíše bývá (hlavně u začátečníků), že do netlistu přidáváme uzly, které tam už nepatří. K pochopení příčin této časté chyby si musíme říct, co to je napěťový uzel a jak se liší od fyzického uzlu (spoje).

Uzel (spoj) je jakékoliv přímé spojení dvou nebo více vodičů. Všechny uzly, které jsou mezi sebou přímo propojeny, takže jsou na stejném potenciálu, tvoří jeden *napěťový* uzel.

Situaci vysvětluje obr. 25. Všechny tři fyzické spoje z obr. 25a) jsou na stejném potenciálu a tvoří tedy jeden napěťový uzel. Počet fyzických spojů je pro funkci nepodstatný, zapojení s jedním spojem podle obr. 25b) bude fungovat úplně stejně jako původní zapojení.



Obr. 25a) Obvod se 3 uzly



Obr. 25b) Ekvivalentní obvod s 1 uzlem

Když budeme ignorovat skutečnost, že na obr. 25a) je vlastně jen jeden napěťový uzel a pojmenujeme každý spoj *jinak* (např. „U1“, „U2“, „U3“), vznikne nám netlist, který bude popisovat obvod se třemi uzly, které už *nebudou mezi sebou spojeny*.

Použijeme-li však *jedno* označení pro všechny tři fyzické spoje (např. „U“), vznikne netlist popisující připojení všech součástek ke *stejnému* napěťovému uzlu, tj. budou vzájemně propojeny.


Nesprávné označení uzlů

```
R1 U1 dalsiuzel1 220k
R2 U2 dalsiuzel2 180k
R3 U3 dalsiuzel3 220k
D1 U3 dalsiuzel KZ241_6V8
```

Správné označení uzlu


```
R1 U dalsiuzel1 220k
R2 U dalsiuzel2 180k
R3 U dalsiuzel3 220k
D1 U dalsiuzel KZ241_6V8
```

K přidávání nadbytečných uzlů nás může přivést nepřehledné zapojení nebo také malá představivost. Typickou ukázkou této druhé možnosti je chyba znázorněná na obr. 26.



Obr. 26. Chyba při identifikaci napěťových uzlů

Obvod je tedy ve SPICE jednoznačně popsán tím, co je zapojeno mezi jednotlivé napěťové uzly. Pro funkčnost zapojení není vůbec podstatné, jak jsou jednotlivé součástky mezi příslušnými uzly „natvarovány“. Na obr. 27 jsou představeny dvě různé grafické podoby Wienova můstku, které mají totožný netlist SPICE a tudíž jsou z hlediska funkce ekvivalentní.



Obr. 27. Dvě ekvivalentní zapojení Wienova můstku

Formát SPICE nás učí správně číst elektrotechnická schémata tím, že nás nutí vyhledávat mezi součástkami podstatné vazby. Užitečným cvičením může být v tomto směru zpětná rekonstrukce schématu z netlistu SPICE. Více o tomto námětu viz literatura [3].

B. TYPOGRAFICKÉ KONVENCE

Jméno jakéhokoliv objektu ve SPICE (součástky, proměnné apod.) musí začínat písmenem a nesmí obsahovat speciální znaky (mezera, hvězdička, otazník apod. – jak je zvykem např. v DOSu). Většina interpretů SPICE nerozlišuje mezi malými a velkými písmeny. Jednotlivá pole v řádku od sebe oddělujeme mezerou, všechny další nadbytečné mezery jsou při zpracování netlistu ignorovány.

Číselná pole mohou být celočíselná (56, -18, ..) nebo reálná s desetinnou *tečkou* (3.141592). Oba dva typy se dají kombinovat s celočíselným exponentem na tzv. vědeckou notaci (5E6, 1.38e-23) nebo pomocí dohodnuté značky na tzv. technickou notaci (36k, .22meg).

Předpona	Násobek	Technická notace	Vědecká notace
Femto	10^{-15}	1f	1E-15
Piko	10^{-12}	1p	1E-12
Nano	10^{-9}	1n	1E-9
Mikro	10^{-6}	1u	1E-6
Mili	10^{-3}	1m	1E-3
Kilo	10^3	1k	1E3
Mega	10^6	1meg	1E6
Giga	10^9	1g	1E9
Tera	10^{12}	1t	1E12


Znaky bezprostředně následující za značkami technické notace jsou ignorovány. Toho se dá využít pro zpřehlednění zápisu, chceme-li např. zadat kapacitu kondenzátoru jako 220nF místo postačujícího 220n. Stejně tak 0.1, .1, 100m, 100mV, 100mVolts jsou ekvivalentní zápisy jedné a téže konstanty.

Je nutno dát pozor na rozdíl mezi mili (m) a mega (meg). Nerozlišuje se mezi malými a velkými písmeny, proto $0.1M \neq 100k$, platí $0.1M = 100u$ a $100k = 0.1meg$.

Nejsou povoleny kombinace typu 3k3, M22 apod. (správně by bylo 3.3k a .22m nebo 220u).

C. SPOLUPRÁCE SPICE – MICRO-CAP


Netlist SPICE je textový soubor a lze jej vytvořit v kterémkoliv textovém editoru. Můžeme jej založit také přímo v editoru Micro-Capu tímto způsobem:



a následně

Obr. 28: Založení netlistu


Zdrojový text SPICE vytvoříme standardním způsobem. Jako ukázka nám poslouží příklad z obr. 2 až 4. Na netlist se pohlíží jako na plnohodnotnou náhradu grafické podoby schématu a okamžitě na něj můžeme aplikovat jakýkoliv druh analýzy. V našem případě spustíme časovou analýzu.



Obr. 29: Spuštění analýzy


Automaticky naběhne menu pro nastavení parametrů analýzy, které vyplníme (viz obr. 30). Budeme chtít vypočítat a zobrazit časové průběhy napětí na zdroji 4,5V, na uzlu 2 proti zemi a na kondenzátoru Cp a proud tekoucí cívkou Ls.

Ještě než analýzu spustíme, uděláme si představu o tom, jaký výsledek bychom měli očekávat. Podívejme se na schéma zapojení na obr. 3. Jakmile se nabije kondenzátor Cp, proud v obvodu se již nebude měnit a cívka se bude chovat jako nulový odpor. V ustáleném stavu tedy bude na cívce nulové napětí a poteče jí proud $4,5V/(1k + 3,3k) = 1,05mA$. Ustálené napětí na kondenzátoru Cp (které se rovná napětí na odporu Rp) by mělo být $1,05mA \times 3,3k = 3,45V$.



Obr. 30: Nastavení parametrů analýzy

Na obr. 31 jsou uvedeny požadované grafické výstupy ze simulátoru Micro-Cap.



Obr. 31: Časová analýza

Vidíme, že napětí na kondenzátoru Cp se opravdu ustálí někde těsně pod hodnotou 3,5V a že proud cívku klesne po počátečním navýšení přibližně na 1mA. Do analýzy křivek by mohla patřit v tomto případě i úvaha o tom, jak je možné, že v uzlu „2“ se objevuje napětí vyšší, než je napětí zdroje.

Podíváme-li se po ukončení simulace na původní netlist, zjistíme, že si do něj simulátor poznačil příkazy, které provedl podle toho, co jsme mu zadali pomocí grafického rozhraní. Při dalším spuštění stejné analýzy se tyto údaje načtou do grafického rozhraní jako editovatelné předvolby.

Chceme-li se naučit něco o syntaxi příkazů pro řízení simulace, je dobré všimnout si těchto změn po každém simulačním běhu. Výskyt nám dosud neznámých příkazů může být příležitostí pro jejich nastudování pomocí dalších experimentů či výkonného „helpu“ programu Micro-Cap.

```

Uvodni priklad
*
*
Vzdroj 1 0 4.5V
Ls 1 2 10mH
Rs 2 3 1k
Rp 3 0 3.3k
Cp 3 0 220nF
*Tady by byly prikazy pro rizeni simulace

.TRAN 2e-005 1M 0 1U UIC
.TEMP 27
.PLOT TRAN V(VZDROJ) V(2) V(CP) 0,5
.PLOT TRAN I(LS) 0,5M
.end
  
```

Obr. 32: Netlist po simulaci

Příkaz

```
.TRAN 2e-005 0.001 0 1e-006 UIC
```

znamená, že bude spuštěna časová (přechodová neboli „transient“) analýza pro simulační čas od 0 do 0,001s (neboli 1ms) s maximálním simulačním krokem 1e-6s (což je 1μs), případný numerický výstup bude škálován po 2e-5s (20 μs). Slovo UIC znamená „Use Initial Conditions“, tzn., že se nebude na začátek simulace počítat pracovní bod (v tomto případě se obvodu „vnutí“ nulové počáteční podmínky, tj. uvažuje se, že kondenzátory jsou zpočátku vybité a cívkami neteče žádný proud).

Příkaz

```
.TEMP 27
```

nastavuje teplotu při běhu simulace na 27°C neboli 300K, což je výchozí teplota ve SPICE.

Příkazy

```
.PLOT TRAN V(VZDROJ) V(2) V(CP) 0,5
```

```
.PLOT TRAN I(LS) 0,5m
```

vykreslují dva grafy jako výsledky časové analýzy. Do prvního grafu se společně umístí časové průběhy napětí na zdroji jménem VZDROJ, napětí uzlu 2 proti zemi a napětí na kondenzátoru Cp, rozsah společné osy „y“ je od 0 do 5V. Druhý graf bude obsahovat časový průběh proudu tekoucího cívku Ls, osa „y“ je od 0 do 5mA.

Práce v programu Micro-Cap je podrobně zmapována v knize [2].

D. HLAVNÍ TYPY ANALÝZ

Stejnoseměrná analýza DC

Je to analýza vhodná pro zjišťování statických charakteristik obvodů. Simulátor přebírá úlohu inteligentního měřiče stejnosměrných charakteristik.

DC analýza probíhá tak, že se opakovaně počítá stejnosměrný pracovní bod pro různé hodnoty napětí nebo proudů stejnosměrných vstupních zdrojů. Některé simulátory dovolují rozšířit analýzu o další vstupní zdroje, např. o symbolické proměnné či teplotu. Na základě těchto údajů může uživatel získat statické nelineární charakteristiky všeho druhu včetně teplotních závislostí.

Časová (přechodová) analýza T

Touto analýzou se zjišťují časové průběhy obvodových veličin. Simulátor se chová jako inteligentní paměťový osciloskop.

Chování obvodu v čase je obecně dáno tím, jak zapojení reaguje na nenulové počáteční podmínky (např. kondenzátor je už zpočátku nabit na nějaké napětí) a jeho odezvou na vnější signály. Pro zjištění počátečních podmínek simulace, tj. napětí a proudů v čase 0, simulátor obvykle počítá stejnosměrný (DC) pracovní bod. Tím se získá základ pro výpočet další odezvy.

Uživatel má většinou k dispozici nástroje pro ovlivňování výpočtu pracovního bodu, např. volbu pro jeho vypnutí. Toho se využívá zvláště tehdy, zajímá-li nás způsob, jakým se obvod dostal do ustáleného stavu (např. náběh do pracovního bodu po zapnutí přístroje, způsob nasazení kmitů oscilátoru apod.).

Frekvenční (kmitočtová) analýza AC

V této analýze se zjišťuje, jak obvod reaguje na buzení sinusovým signálem malé amplitudy v zadaném rozsahu frekvencí. Simulátor se chová jako frekvenční analyzátor.

Simulátor nejprve zjistí stejnosměrný pracovní bod (tj. klidový stav) a pro tento pracovní bod vypočítá linearizované parametry obvodu. Pak nahradí určené signálové zdroje (to je dáno polem AC v modelu těchto zdrojů, viz kap. 4) zdroji sinusového signálu (amplituda a fáze takového zdroje jsou zadávány jako parametry za polem AC) a proládňuje jejich frekvenci v zadaném rozsahu. Pro každou frekvenci počítá ustálené hodnoty amplitudy a fáze sinusových obvodových veličin. Na základě těchto hodnot může uživatel získat nejrůznější frekvenční charakteristiky (klasické amplitudové a fázové, vstupní a výstupní impedance a admitance, charakteristiky v komplexní rovině apod.).

Ve formátu SPICE je implementována podpora dalších typů analýz, se kterými pracuje i většina moderních simulátorů. Více je možno nastudovat ve skriptech [1] a v knize [2].

LITERATURA

- [1] Kejhar, M., Kirschner, M., Musil, V., Stříbrný, V.: Program SPICE v příkladech. Skriptum. Vydavatelství ČVUT, Praha 1995.
- [2] Biolk, D.: Řešíme elektronické obvody aneb kniha o jejich analýze. BEN – technická literatura, ISBN 80-7300-125-X, 2004.
- [3] Biolk, Z.: SPICE jako pomůcka k porozumění schématu. Perspektivy elektroniky, Rožnov p.R., 2003. K dispozici na <http://www.roznovskastredni.cz/dwnl/pel2003/3/Spice.pdf> .
- [4] Biolk, Z.: Počítačové experimenty s podporou SPICE. Perspektivy elektroniky, Rožnov p.R., 2005. K dispozici na <http://www.roznovskastredni.cz/dwnl/pel2005/10/biolk.pdf> .

© Ing. Zdeněk Bielek, Ph.D., 2009

Úvod do SPICE pomocí programu Micro-Cap

Učební texty

Pro potřebu žáků SŠIEŘ Rožnov p.R. v rámci předmětu PRV4